Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematical Sciences
goutzicj@oneonta.edu

Lesson 6
The Substitution Rule

Initializations

```
> restart;
  with(student):
>
```

6.1 The Substitution Rule

Examples

Example 6.1.1

Compute $\int x\sqrt{1+5x^2}\,dx$.

Solution
First code the integral.

```
> e1:=Int(x*sqrt(1+5*x^2), x);
```

$$e1 := \int x\sqrt{1+5x^2}\,dx \quad \textbf{(2.1.1.1)}$$

It is usually advantageous to code the actual substitution as a separate variable. The latter allows for easy translation later in the solution process.

```
> rr:=u=1+5*x^2;
```

$$rr := u = 1+5x^2 \quad \textbf{(2.1.1.2)}$$

Make the substitution.

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int \frac{1}{10}\sqrt{u}\,du \quad \textbf{(2.1.1.3)}$$

Evaluate this elementary integral.

```
> e3:=value(e2)+c;
```

$$e3 := \frac{1}{15}u^{3/2}+c \quad \textbf{(2.1.1.4)}$$

Translate the result into terms of *x*.

```
> answer:=subs(rr, e3);
```

$$answer := \frac{1}{15}\left(1+5x^2\right)^{3/2}+c \quad \textbf{(2.1.1.5)}$$

```
>
```

Example 6.1.2

Evaluate the integral $\int_0^{\pi} 4\,x^2 \sin 7x^3\,dx$.

Solution

Substitutions work for definite as well as for indefinite integrals. Here we replace the argument $7x^3$ of the sine function by the new variable u.

```
> e1:=Int(4*x^2*sin(7*x^3), x=0..Pi);
```

$$e1 := \int_0^{\pi} 4\,x^2 \sin\left(7\,x^3\right)\,\mathrm{d}x \qquad (2.1.2.1)$$

```
> rr:=u=7*x^3;
```

$$rr := u = 7\,x^3 \qquad (2.1.2.2)$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int_0^{7\,\pi^3} \frac{4}{21} \sin(u)\,\mathrm{d}u \qquad (2.1.2.3)$$

Observe how the **changevar** command automatically adjusts the integration limits.

```
> e3:=value(e2);
```

$$e3 := \frac{4}{21} - \frac{256}{21}\cos\left(\pi^3\right)^7 + \frac{64}{3}\cos\left(\pi^3\right)^5 - \frac{32}{3}\cos\left(\pi^3\right)^3 + \frac{4}{3}\cos\left(\pi^3\right) \qquad (2.1.2.4)$$

The **combine** command with the **trig** option will cut this formula down to size.

```
> answer[simplified]:=combine(e3, trig);
  evalf(answer[simplified]);
```

$$answer_{simplified} := \frac{4}{21} - \frac{4}{21}\cos\left(7\,\pi^3\right)$$

$$0.3738446548 \qquad (2.1.2.5)$$

```
>
```